Bivariate Correlation and Multiple Regression Analyses for Continuous Variables Using SAS
(commands=finan_regression.sas)
/**************************************/

/* BIVARIATE CORRELATION ANALYSIS FOR */

/* TWO CONTINUOUS VARIABLES IN SAS    */

/**************************************/

/* INDICATE LIBRARY CONTAINING PERMANENT SAS DATA SET "CARS" */

libname sasdata2 V9 "C:\temp\sasdata2";

First, we consider commands to generate scatter plots. 

In INSIGHT: go to the command dialog box and type “INSIGHT”, without the quotes. Click on Scatter Plot (Y,X) and select MPG as Y and Year or Weight as X.
goptions reset=all;

goptions device=win;

proc gplot data = sasdata2.cars;

  plot mpg*year;

  plot mpg*weight;

  symbol value=dot;

run;

quit;

/* ARE THE RELATIONSHIPS LINEAR? */

/* INVESTIGATE STRANGE OBSERVATION */

proc print data = sasdata2.cars;

  where year eq 0;

run;
        Obs     MPG    ENGINE    HORSE    WEIGHT    ACCEL    YEAR    ORIGIN    CYLINDER

         35       9        4        93      732        9       0       .          .

/* REMOVE STRANGE OBSERVATION WITH YEAR = 0, AND INVESTIGATE SCATTERPLOT AGAIN. */

data cars2;

   set sasdata2.cars;

   if year ne 0;

run;

proc gplot data = cars2;

  plot mpg*year;

  symbol value=dot;

run; quit;

Calculate Pearson correlation coefficients for the variables of interest. 

In INSIGHT: select Multivariate(YX), and all variables will be “Y” variables.
proc corr data = cars2;

  var weight year mpg;

run;
                               Pearson Correlation Coefficients

                                  Prob > |r| under H0: Rho=0

                                    Number of Observations

                                              WEIGHT          YEAR           MPG

               WEIGHT                        1.00000      -0.30990      -0.83014

               Vehicle Weight (lbs.)                        <.0001        <.0001

                                                 405           405           397

               YEAR                         -0.30990       1.00000       0.57608

               Model Year (modulo 100)        <.0001                      <.0001

                                                 405           405           397

               MPG                          -0.83014       0.57608       1.00000

               Miles per Gallon               <.0001        <.0001

                                                 397           397           397

Consider the nomiss and spearman options. Nomiss is for listwise deletion of missing values (as opposed to the default option of pairwise deletion), while Spearman is a non-parametric test of correlation (Pearson correlation assumes normality).
proc corr data = cars2 nomiss;

  var weight year mpg;

run;

                         Pearson Correlation Coefficients, N = 397

                                  Prob > |r| under H0: Rho=0

                                              WEIGHT          YEAR           MPG

               WEIGHT                        1.00000      -0.30067      -0.83014

               Vehicle Weight (lbs.)                        <.0001        <.0001

               YEAR                         -0.30067       1.00000       0.57608

               Model Year (modulo 100)        <.0001                      <.0001

               MPG                          -0.83014       0.57608       1.00000

               Miles per Gallon               <.0001        <.0001

proc corr data = cars2 spearman;

  var weight year mpg;

run;

                             Spearman Correlation Coefficients

                                  Prob > |r| under H0: Rho=0

                                    Number of Observations

                                              WEIGHT          YEAR           MPG

               WEIGHT                        1.00000      -0.28281      -0.87402

               Vehicle Weight (lbs.)                        <.0001        <.0001

                                                 405           405           397

               YEAR                         -0.28281       1.00000       0.57053

               Model Year (modulo 100)        <.0001                      <.0001

                                                 405           405           397

               MPG                          -0.87402       0.57053       1.00000

               Miles per Gallon               <.0001        <.0001

                                                 397           397           397

************************************/

/* MULTIPLE REGRESSION ANALYSIS FOR */

/* CONTINUOUS VARIABLES IN SAS      */

/************************************/

Fit a multiple regression model to the CARS data, where MPG is the dependent variable, and WEIGHT and YEAR are the continuous predictor variables. Generate a plot of studentized residuals (RSTUDENT, based on the model fitted by deleting the current observation from the data set), versus the predicted values, and output the predicted values and residuals to a new SAS data set, REGDAT..
proc reg data = cars2;

  model mpg = weight year / clb;

  plot rstudent.*predicted.;

  output out=regdat p=predict r=resid rstudent=rstudent;

run;

quit;
                                      The REG Procedure

                                         Model: MODEL1

                           Dependent Variable: MPG Miles per Gallon

                    Number of Observations Read                        405

                    Number of Observations Used                        397

                    Number of Observations with Missing Values           8

                                     Analysis of Variance

                                            Sum of           Mean

        Source                   DF        Squares         Square    F Value    Pr > F

        Model                     2          19385     9692.36159     820.07    <.0001

        Error                   394     4656.64882       11.81891

        Corrected Total         396          24041

                     Root MSE              3.43786    R-Square     0.8063

                     Dependent Mean       23.55113    Adj R-Sq     0.8053

                     Coeff Var            14.59744

                                      Parameter Estimates

                                                Parameter       Standard

Variable     Label                      DF       Estimate          Error    t Value    Pr > |t|

Intercept    Intercept                   1      -14.27636        3.97422      -3.59      0.0004

WEIGHT       Vehicle Weight (lbs.)       1       -0.00667     0.00021481     -31.07      <.0001

YEAR         Model Year (modulo 100)     1        0.75791        0.04909      15.44      <.0001

                                     Parameter Estimates

           Variable     Label                      DF       95% Confidence Limits

           Intercept    Intercept                   1      -22.08968       -6.46304

           WEIGHT       Vehicle Weight (lbs.)       1       -0.00710       -0.00625

           YEAR         Model Year (modulo 100)     1        0.66140        0.85442

/* WHY IS THERE A DIAGONAL LINE AT THE BOTTOM OF THE FITTED-RESIDUAL PLOT? */

/* CHECK THE RESIDUAL DISTRIBUTION FOR NORMALITY USING THE SAVED DATA SET */
proc univariate data=regdat noprint;

  var rstudent;

  histogram / normal;

  probplot / normal (mu=est sigma=est);

run;

                                  The UNIVARIATE Procedure

                               Fitted Distribution for rstudent

                               Parameters for Normal Distribution

                                 Parameter   Symbol   Estimate

                                 Mean        Mu       0.001521

                                 Std Dev     Sigma    1.005856

                        Goodness-of-Fit Tests for Normal Distribution

                  Test                  ---Statistic----   -----p Value-----

                  Kolmogorov-Smirnov    D     0.04323989   Pr > D      0.071

                  Cramer-von Mises      W-Sq  0.16822812   Pr > W-Sq   0.015

                  Anderson-Darling      A-Sq  1.55291383   Pr > A-Sq  <0.005

/* CALCULATE MEANS ON PREDICTORS */
proc means data = cars2;

   var weight year;

run;

                                     The MEANS Procedure

Variable  Label                      N          Mean       Std Dev       Minimum       Maximum

ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

WEIGHT    Vehicle Weight (lbs.)    405       2975.09   843.5463681       1613.00       5140.00

YEAR      Model Year (modulo 100)  405    75.9358025     3.7417668    70.0000000    82.0000000

ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ 

/* CREATE NEW VARIABLES */

data cars3;

   set cars2;

   logmpg = log(mpg);

   wgtcent = weight - 2975.09; /* 2975.09 = mean of weight */

   wgtcent2 = wgtcent**2;

   yearcent = year - 75.94; /* 75.94 = mean of year */

run;

/* REFIT THE MODEL USING THE NEW VARIABLES. */

proc reg data = cars3;

  model logmpg = wgtcent wgtcent2 yearcent / clb;

  plot rstudent.*predicted.;

  output out=regdat2 p=predict r=resid rstudent=rstudent;

run;

quit;
                                      The REG Procedure

                                         Model: MODEL1

                                  Dependent Variable: logmpg

                    Number of Observations Read                        405

                    Number of Observations Used                        397

                    Number of Observations with Missing Values           8

                                     Analysis of Variance

                                            Sum of           Mean

        Source                   DF        Squares         Square    F Value    Pr > F

        Model                     3       39.66793       13.22264     978.08    <.0001

        Error                   393        5.31296        0.01352

        Corrected Total         396       44.98090

                     Root MSE              0.11627    R-Square     0.8819

                     Dependent Mean        3.10366    Adj R-Sq     0.8810

                     Coeff Var             3.74626

                                      Parameter Estimates

                     Parameter      Standard

  Variable    DF      Estimate         Error   t Value   Pr > |t|     95% Confidence Limits

  Intercept    1       3.05880       0.00849    360.39     <.0001       3.04211       3.07548

  wgtcent      1   -0.00032989    0.00000800    -41.24     <.0001   -0.00034562   -0.00031416

  wgtcent2     1   5.507953E-8   8.612069E-9      6.40     <.0001   3.814805E-8   7.201102E-8

  yearcent     1       0.03274       0.00168     19.49     <.0001       0.02944       0.03604

/* DO THE DIAGNOSTIC PLOTS LOOK BETTER? */

proc univariate data=regdat2 noprint;

  var rstudent;

  histogram / normal;

  probplot / normal (mu=est sigma=est);

run;

/* USE PROC TRANSREG TO SEE IF THERE IS A BETTER TRANSFORMATION of Y. */

proc transreg data = cars3;

   model boxcox(mpg) = identity(wgtcent wgtcent2 yearcent);

run;

                                   The TRANSREG Procedure

                                  Transformation Information

                                        for BoxCox(MPG)

                               Lambda      R-Square    Log Like

                                -3.00          0.77    -725.478

                                -2.75          0.79    -676.705

                                -2.50          0.81    -630.006

                                -2.25          0.82    -585.701

                                -2.00          0.84    -544.178

                                -1.75          0.85    -505.904

                                -1.50          0.86    -471.426

                                -1.25          0.87    -441.362

                                -1.00          0.88    -416.365

                                -0.75          0.88    -397.072

                                -0.50          0.88    -384.025

                                -0.25          0.88    -377.588 <

                                 0.00 +        0.88    -377.877 *

                                 0.25          0.88    -384.738

                                 0.50          0.87    -397.767

                                 0.75          0.86    -416.383

                                 1.00          0.85    -439.907

                                 1.25          0.84    -467.646

                                 1.50          0.82    -498.953

                                 1.75          0.80    -533.254

                                 2.00          0.78    -570.068

                                 2.25          0.76    -608.999

                                 2.50          0.74    -649.730

                                 2.75          0.72    -692.008

                                 3.00          0.69    -735.635

                             < - Best Lambda

                             * - Confidence Interval

                             + - Convenient Lambda

/* FOR BOX-COX TRANSFORMATIONS, Z = (Y^L - 1) / L. */

/* L = 0 => log transformation. */

/* INCLUSION OF ADDITIONAL PREDICTORS / EXCLUSION OF OUTLIERS */

/* MAY IMPROVE THE FIT AND THE DIAGNOSTIC PLOTS EVEN MORE. */

/* EXAMINE THE FIT IN MORE DETAIL USING SAS INSIGHT. */

/* FIT THE MODEL USING ANALYZE -> FIT (YX), */

/* AND EXAMINE THE FIT USING GRAPHS. */

/* PREDICTION INTERVALS ARE AVAILABLE FOR SIMPLE LINEAR */

/* REGRESSION MODELS WHEN USING INSIGHT. */

